MQL4 Book 2

Common Functions

One of the most widely used functions is Comment().

Comment() Function

void Comment(...)
This function introduces a comment defined by a user into the upper left corner of a chart window. Parameters can be of any type. Number of parameters cannot exceed 64. Arrays cannot be passed to the Comment() function. Arrays should be introduced elementwise. Data of double type are written with 4 digits after the decimal point. For showing figures with accuracy use the DoubleToStr() function. Bool, datetime and color types will be written as digits. To show data of datetime type as a string use the TimeToStr() function.

Parameters:
... - any values separated by commas.

Example of using this function can be a simple EA comment.mq4 reflecting the information about the number of orders.

//--------------------------------------------------------------------
// comment.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
int start()                            // Special function start
  {
   int Orders=OrdersTotal();           // Number of orders
   if (Orders==0)                      // If numb. of ord. = 0
      Comment("No orders");            // Comment to the window corner
   else                                // If there are orders
      Comment("Available ",Orders," orders." );// Comment
   return;                             // Exit 
  }
//--------------------------------------------------------------------
At the program beginning the total number of orders is counted by the OrdersTotal() function. If the Orders variable (number of orders) is equal to 0, the Comment() function with "No orders" parameter is executed. If there is at least one order, Comment() with a list of parameters comma separated will be executed. In this case 3 parameters are used: the first one is a string value "Available", second is an integer value Orders and the third one is a string value "orders.". As a result of the function execution at each start of the special function start() one of the messages will be shown in the upper left corner of a chart window. Fig. 132 illustrates a chart window in the situation when there is one order present.


For the reproduction of sound files PlaySound() function is used.

PlaySound() Function

void PlaySound(string filename)
The function plays a sound file. The file must be located in terminal_directory\sounds or its subdirectory.

Parameters:
filename - path to a sound file.

A set of recommended sound files can be found in the attachment - Sound Files.

In some cases a program can be written to support a dialog with a user. Function MessageBox() is used for this purpose.

MessageBox() function

int MessageBox(string text=NULL, string caption=NULL, int flags=EMPTY)
MessageBox function creates and displays a message box, it is also used to manage the dialog window. A message box contains a message and header defined ina program, as well as any combination of predefined icons and push buttons. If a function is executed successfully, the returned value is one of the return code values of MessageBox(). The function cannot be called from a custom indicator, because indicators are executed in the interface thread and may not decelerate it.

Parameters:
text - a text containing a message to be displayed;
caption - an optional text to be displayed in the message box. If the parameter is empty, an EA name will be displayed in the box header;
flags - optional flags defining the type and behavior of the dialog box. Flags can be a combination of flags from flag groups (see MessageBox Return Codes).

Let us consider an example of MessageBox() usage.

	[image: image1.png]



	Problem 31. Write a code of an EA displaying a message box with a question of closing all orders 5 minutes prior to important news release. If a user clicks Yes, all orders should be closed, if No is pushed, no actions should be performed. 


The EA supporting a dialog with a user (dialogue.mq4) can have the following code:

//--------------------------------------------------------------------
// dialogue.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------- 1 --
#include <WinUser32.mqh>               // Needed to MessageBox
extern double Time_News=15.30;         // Time of important news
bool Question=false;                   // Flag (question is not put yet)
//--------------------------------------------------------------- 2 --
int start()                            // Special function start
  {
   PlaySound("tick.wav");              // At each tick
   double Time_cur=Hour()+ Minute()/100.0;// Current time (double)
   if (OrdersTotal()>0 && Question==false && Time_cur<=Time_News-0.05)
     {                                 // Providing some conditions
      PlaySound("news.wav");           // At each tick
      Question=true;                   // Flag (question is already put)
      int ret=MessageBox("Time of important news release. Close all orders?",
      "Question", MB_YESNO|MB_ICONQUESTION|MB_TOPMOST); // Message box
      //--------------------------------------------------------- 3 --
      if(ret==IDYES)                   // If the answer is Yes
         Close_Orders();               // Close all orders
     }
   return;                             // Exit 
  }
//--------------------------------------------------------------- 4 --
void Close_Orders()                    // Cust. funct. for closing orders
  {
   Alert("Function of closing all orders is being executed.");// For illustration
   return;                             // Exit 
  }
//--------------------------------------------------------------- 5 --
In the block 1-2 WinUser32.mqh file is included into the program; in this file MessageBox() return codes are defined. Also in this block the external variable Time_news is introduced - this is the time of important news release. During the whole period of EA execution a question about closing orders should be displayed only once. To trace whether the question has been already displayed in the EA 'Question' variable is declared.

At each start of the special function start() (block 2-3) PlaySound() is executed. The played tick.wav sound resembles a weak click that denotes the best way the fact of a new tick appearance. The decision about using sound in a program is made by a programmer. In some cases it is very useful to use sounds. For example, a sound may denote the fact of an EA execution. Other sounds may correspond to other events, for example triggering of a trading criterion, order closing, etc.

Value of the actual variable Time_cur corresponds to the current server time. In the EA conditions, at which the message box must be displayed, are analyzed. If there is one or several orders, the message box has not been shown yet and the server time differs from important news release time by less than 5 minutes, certain actions are performed in the program. First of all function PlaySound() is executed, the played sound attracts the attention of a user. The Question flag gets the true value (do not show the next time). In the next line MessageBox() is executed:

      int ret=MessageBox("Time of important news release. Close all orders?",
      "Question", MB_YESNO|MB_ICONQUESTION|MB_TOPMOST); // Message box
In this case the value of a string constant "Time of important news release. Close all orders?" will be displayed in a dialog box, and "Question" value will be reflected in the upper box line. The MB_YESNO flag denotes the presence of buttons - in this case Yes and No buttons (see MessageBox Return Values). The MB_ICONQUESTION flag defines an icon displayed in the left part of the message box (each operating environment has its own set of icons, Fig. 133 illustrates an icon from Windows XP set). The MB_TOPMOST flag provides the box with the property "always on top", i.e. the box will be always visible irrespective of what programs are being executed at the moment on the computer. As a result of the execution of MessageBox() with indicated parameters a message box is displayed:

Fig. 133. Dialog box displayed as a result of MessageBox() execution.

At the moment when the message box is shown the program execution is held up until a user clicks on a button in the message box. As soon as it happens, control will be passed to the line following MessageBox() call, in this case to the block 3-4. This property of a message box to hold control is very important and it must be taken into account at a program development. For example, if a user left his computer and a message box was shown at this moment, during the whole time when a user is absent (until a button is pressed), the program will be waiting for response and no code will be executed in this period.

Note, before a message box is displayed the program execution is accompanied by a sound of ticks. When the message box is displayed another sound is played. In the period when the dialog box is open and waiting for response no sound is played which illustrates the fact of control holding while the dialog box is open. After a button is pressed, the program execution will continue and the sound of ticks will be played again.

If a user clicks Yes, the Close_Orders() function will be called; this function is used for closing orders. In this example the function contents is not described; to denote its execution the Alert function is executed ("Function of closing all orders is being executed."). If a user clicks No, the function of closing orders is not called. In the current session of the EA execution the message box will not be shown again.

Common Functions

	Function
	Summary Info

	Alert
	Displays a message box containing user-defined data. Parameters may be of any type. Number of parameters cannot exceed 64.

	Comment
	Displays a comment defined by a user in the upper loft corner of a chart window. Parameters may be of any type. Number of parameters cannot exceed 64.

	GetTickCount
	GetTickCount() returns number of milliseconds passed since a system was started. The counter is limited by the resolution of the system timer. Since time is stored as an unsigned integer, it is overfilled every 49.7 days.

	MarketInfo
	Returns various information about securities listed in "Market watch" window. Part of the information about the current security is stored in predefined variables (see MarketInfo() Identifiers).

	MessageBox
	MessageBox function creates and displays a message box, it is also used to manage the dialog window. A message box contains a message and header defined ina program, as well as any combination of predefined icons and push buttons. If a function is executed successfully, the returned value is one of the return code values of MessageBox(). The function cannot be called from a custom indicator, because indicators are executed in the interface thread and may not decelerate it.

	PlaySound
	Plays a sound file. The file must be located in the terminal_dir\sounds directory or in its subdirectory.

	Print
	Prints a message to the experts log. Parameters can be of any type. Amount of passed parameters cannot exceed 64.

	SendFTP
	Sends a file to the address specified in setting window of "Publisher" tab. If the attempt fails, it retuns FALSE. The function does not operate in the testing mode. This function cannot be called from custom indicators, either. The file to be sent must be stored in the terminal_directory\experts\files folder or in its sub-folders. It will not be sent if there is no FTP address and/or access password specified in settings.

	SendMail
	Sends an email to the address specified in settings window of "Email" tab. The sending can be disabled in settings, or it can be omitted to specify the e-mail address.

	Sleep
	The Sleep() function suspends execution of the current expert within the specified interval. Sleep() cannot be called from custom indicators, because indicators are executed in the interface thread and may not decelerate it.
The checking of the expert stop flag status every 0.1 second is built into the function.


For the detailed description of these and other functions please refer to Documentation at MQL4.community, MetaQuotes Software Corp. website or to "Help" section in MetaEditor.

Graphical Objects

Graphical object is an image in the symbol window; it can be selected, moved, modified or deleted.

Graphical objects include, for example, horizontal and vertical lines, linear regression channel, Fibonacci levels, rectangle, text mark, etc. Such images as indicator lines, indicator levels, candlesticks, comments written by the Comment() function and others cannot be selected and deleted, that is why they do not belong to graphical objects.

Graphical object are drawn by the client terminal in a security window in accordance with preset coordinates. Each graphical object depending on its type has one, two or three coordinates and other adjustable parameters. Any graphical object can be placed in a chart window manually (from the toolbar of a system menu), and also as a result of the execution of an application program started in the same window, including an Expert Advisor, script or custom indicator. Type and location of a graphical object can be modified manually or by a program sending new values of coordinates and other parameters to a graphical object.

Ways of Positioning Graphical Objects

There are two ways of positioning objects accepted in MQL4: relative to a chart and relative to a security window. To illustrate the difference between these methods, let us place manually two objects in a security window: text (OBJ_TEXT) and a text mark (OBJ_LABEL). We can use A and T buttons from the toolbar of the client terminal. Let us set the window size so that it is equal to half of screen size (Fig. 134). Let us see how these graphical objects will react to the window size changes (as well as to the horizontal and vertical scaling of the price chart).

Fig. 134. Graphical objects with different methods of positioning in a security window.

Positioning Relative to a Chart Window

The graphical object OBJ_LABEL will remain immovable if a window size is changed by way of shifting its right or lower borders. But if the window size is changed by shifting its upper or lower border, the object will be also shifted, though the position of the object relative to these borders will remain unchanged. This happens because OBJ_LABEL is positioned relative to security window borders. In this case the reference point of the graphical object to a security window is the upper left corner of a chart6 window. Coordinates of the object relative to the indicated point are set in pixels - 193 and 48 (Fig. 135).

Fig. 135. Settings of the graphical object OBJ_LABEL.

The reference point of the object coordinates (in this case) is the upper left corner of a cursor frame visible when selected by a mouse. In the upper left corner of the cursor frame you can see a small point indicating the settings of this graphical object. If another reference point is indicated, the point in the cursor frame will be indicated in another corner.

When new bars appear in a chart window, an object like OBJ_LABEL will remain immovable in the window. Using of this object is convenient if it is necessary to display text information of general character, for example, information about termination of trading, value of a limiting distance changed by a broker, etc.

Positioning Relative to a Chart

At any method of windows size changing, as well as at chart scaling, an object of OBJ_TEXT type does not change its position relative to a chart. The reference point of such an object is the middle of the upper line of a cursor frame, its X coordinate is time, Y coordinate is a security price (Fig. 136).

Fig. 136. Settings of the graphical object OBJ_TEXT.

As new bars appear in a chart window, the position of OBJ_TEXT does not change relative to a chart, i.e. with the appearance of new bars the object will be shifted to the left together with the chart; and when there will be enough bars, the object will move further to the left out of the window borders.

This or that method of positioning the own property of a certain object type and cannot be changed by a user, even in a program way. The majority of graphical objects is positioned relative to a chart, i.e. in time and price coordinates.

Creating Graphical Objects and Changing Their Properties

To create a graphical object means to place in a chart window one of objects of predefined types (see Types and Properties of Graphical Objects). For object creation the following function is used:

ObjectCreate() Function

bool ObjectCreate(string name, int type, int window, datetime time1, double price1, datetime time2=0, 
double price2=0, datetime time3=0, double price3=0)
The function creates an object of an indicated type with a preset name and coordinates in the indicated chart subwindow. Number of the object coordinates can be from 1 to 3 depending on the object type. If an object is successfully created, the function returns TRUE, otherwise FALSE. To get additional information about an error call the GetLastError() function.

Coordinates must be passed in pairs - time and price. For example OBJ_VLINE needs only time, but price should also be passed (any value). Graphical object of OBJ_LABEL type ignores coordinates specified in the function; to set OBJPROP_XDISTANCE and OBJPROP_YDISTANCE of this object the ObjectSet() function must be used.

Parameters:

· name - object name; 

· type - object type (can be one of predefined object types); 

· window - number of window into which an object will be added. Numeration of chart subwindows (if there are subwindows with indicators present) starts from 1, the number of the main window is always 0; the indicated widow number must be larger than or equal to 0 and less than the value returned by the WindowsTotal() function; 

· time1 - time of the first coordinate; 

· price1 - price of the first coordinate; 

· time2 - time of the second coordinate; 

· price2 - price of the second coordinate; 

· time3 - time of the third coordinate; 

· price3 - price of the third coordinate. 

Each graphical object has some (peculiar to it) adjustable parameters. For example, besides defined coordinates, you can specify color, message text (for some objects), line styles (for other objects), etc. For changing properties use the following function:

ObjectSet() Function

bool ObjectSet(string name, int prop_id, double value)
The function changes the value of the indicated object property. In case of success the function returns TRUE, otherwise FALSE. To get the error information call the GetLastError() function.

Parameters:

· name - object name; 

· prop_id - object properties identifier (one of object properties is indicated); 

· value - new value of the indicated property. 

All graphical objects may have a text description. The text description of each object is available to a user and can be changed from an object properties toolbar or in a programmed way. For OBJ_TEXT and OBJ_LABEL this description is their main contents and and is always displayed as a text line, text descriptions of other objects are displayed near the object if the option "Show object descriptions" is enabled in a symbol properties window (F8). To change the text description the following function is used:

ObjectSetText() Function

bool ObjectSetText(string name, string text, int font_size, string font_name=NULL, color text_color=CLR_NONE)
The function is used for changing an object description. In case of success TRUE is returned, otherwise - FALSE. To get the error information call the GetLastError() function. Parameters font_size, font_name and text_color are used only for OBJ_TEXT and OBJ_LABEL. For objects of other types these parameters are ignored.

Parameters:

· name - object name; 

· text - object description text; 

· font_size - font size in points; 

· font_name - font name; 

· text_color - text color. 

Lets us analyze an example of an Expert Advisor,in which functions of managing graphical objects are used.

	Problem 32. Using a graphical object inform a user about trading criteria defined on the basis of MACD values.


MACD is very often used by traders for the formation of trading criteria. The indicator is represented by two lines - main and signal. A trading criteria is considered to be realized when the lines cross. If the main indicator line (usually gray histogram) crosses the signal line (usually red dotted line) downwards, this is a signal to sell, id upwards - to buy. In intervals between line crossing market orders should be held open, and when a contrary criterion triggers, the orders should be closed and opposite once opened. Thus four message types should be prepared: opening of Buy, opening of Sell, holding of Buy, holding of Sell.

In this problem all messages are mutually exclusive, i.e. the situation when two or more messages should be shown is impossible. That is why in this case one graphical object can be used; the object will be always present on th screen, but it will be changed from time to time. Let us draw this object in the upper right corner of the window, in which the EA will operate. Since the object position should not be changed, it is convenient to use an object of OBJ_LABEL type, because it is positioned relative to a chart window.

As a solution of Problem 32 let us view the EA grafobjects.mq4 using the graphical object OBJ_LABEL:

//--------------------------------------------------------------------
// grafobjects.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
int start()                            // Special function start
  {
//--------------------------------------------------------------- 1 --
   int Sit;
   double MACD_M_0,MACD_M_1,           // Main line, 0 and 1st bar
   MACD_S_0,MACD_S_1;                  // Signal line, 0 and 1st bar
   string Text[4];                     // Declaring a string array
   color  Color[4];                    // Declaring an array of colors
 
   Text[0]= "Opening of Buy";          // Text for different situations
   Text[1]= "Opening of Sell";
   Text[2]= "Holding of Buy";
   Text[3]= "Holding of Sell";
 
   Color[0]= DeepSkyBlue;              // Object color ..
   Color[1]= LightPink;                // .. for different situations
   Color[2]= Yellow;
   Color[3]= Yellow;
//--------------------------------------------------------------- 2 --
   ObjectCreate("Label_Obj_MACD", OBJ_LABEL, 0, 0, 0);// Creating obj.
   ObjectSet("Label_Obj_MACD", OBJPROP_CORNER, 1);    // Reference corner
   ObjectSet("Label_Obj_MACD", OBJPROP_XDISTANCE, 10);// X coordinate
   ObjectSet("Label_Obj_MACD", OBJPROP_YDISTANCE, 15);// Y coordinate
//--------------------------------------------------------------- 3 --
   MACD_M_0 =iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_MAIN,0);  // 0 bar
   MACD_S_0 =iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_SIGNAL,0);// 0 bar
   MACD_M_1 =iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_MAIN,1);  // 1 bar
   MACD_S_1 =iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_SIGNAL,1);// 1 bar
//--------------------------------------------------------------- 4 --
   // Analyzing situation 
   if(MACD_M_1=MACD_S_0)                      // Crossing upwards
      Sit=0;
   if(MACD_M_1>MACD_S_1 && MACD_M_0<=MACD_S_0)// Crossing downwards
      Sit=1;
   if(MACD_M_1>MACD_S_1 && MACD_M_0>MACD_S_0) // Main above signal
      Sit=2;
   if(MACD_M_1
In the EA block 1-2 parameters are defined, in particular element values of Text[] and Color[] are set. Further they are used for changing object properties. In the block 2-3 the object is created and values of some its properties are set. Let us analyze this block in details. According to this EA code line a graphical object is created in the window, in which the EA is executed:

   ObjectCreate("Label_Obj_MACD", OBJ_LABEL, 0, 0, 0);// Creating obj.
"Label_Obj_MACD" value denotes that this name is assigned to the object (a name is assigned to an object by a programmer in his own discretion). OBJ_LABEL - is the object type identifier; it denotes that the created object will be of exactly this type (chosen from the list of possible types). The first of the next three zeros denotes that the object is created in the main window (the main window where the chart is displayed, always has the index 0).

The next two zeros set coordinates for the created object. According to this coordinated the object will be drawn in the indicated window. In this case the created OBJ_LABEL does not use time and price coordinates. Please note that in OjectCreate() description only time and price coordinates are specified. Moreover, coordinates of the second and the third pairs have default values, while there are no default values for the first pair of coordinates. It means that though OBJ_LABEL does not need time and price coordinates at all, some values must be specified in ObjectCreate() function call. In this case zeros are indicated, though any other values can be written - anyway these values will be neglected during the setup of OBJ_LABEL properties.

In the next three lines some property values are set to the earlier created object named Label_Obj_MACD:

   ObjectSet("Label_Obj_MACD", OBJPROP_CORNER, 1);    // Reference corner
   ObjectSet("Label_Obj_MACD", OBJPROP_XDISTANCE, 10);// X coordinate
   ObjectSet("Label_Obj_MACD", OBJPROP_YDISTANCE, 15);// Y coordinate
For the reference corner (OBJPROP_CORNER) 1 is set, which means the upper right corner of the earlier defined main window. In the next two lines distances from the object to a reference corner are set on pixels: horizontal distance (OBJPROP_XDISTANCE) 10 pixels and vertical distance (OBJPROP_YDISTANCE) 15 pixels. At this program execution stage the object is already created, has its unique name and defined main properties.

To make the object show a necessary text, first we need to calculate what this text should look like. For this purpose first in block 3-4 the position of MACD lines is detected on the current and previous bars, then in block 4-5 Sit value corresponding to the current situation is calculated (see also Fig. 107 and callstohastic.mq4)

In the next line object properties depending on the current situation are defined:

                                       // Changing object properties
   ObjectSetText("Label_Obj_MACD",Text[Sit],10,"Arial",Color[Sit]);
As a result of ObjectSetText() execution a text description is assigned to the object named Label_Obj_MACD - the value of the string variable Text[Sit]. This value will be different for different situations depending on values of Sit variable. For example, if the main line crosses the signal one downwards, in block 4-5 Sit gets the value 1, as a result the graphical object will get the text description contained in the Text[1] array element, i.e. "Opening of Sell". Other parameters: 10, "Arial" and Color[Sit] denote font size, name and color for the text description.

As a result of the EA execution the following will appear in EURUSD window:

Fig. 137. Result of the EA grafobjects.mq4 operation at the moment when criterion to sell triggers.

In Fig. 137 there is a main window and MACD subwindow. It should be noted here that for a normal EA operation presence of this indicator on the symbol window is not necessary, because trading criteria in the EA are calculated as a result of a technical indicator function execution which is not connected with the indicator displaying. Here the indicator is shown only for visual explanation of the moment of a trading criterion triggering when the necessary text description of the graphical object is shown. The EA will operate in the similar way at all other combinations of the mutual position of indicator lines each time showing a description corresponding to a situation.

Deleting Graphical Objects

The analyzed Expert Advisor grafobjects.mq4 has a small disadvantage. After the EA stops operating, a graphical object will remain in the chart window (his properties will remain the same as at t he moment of its last change). Graphical objects are not deleted automatically. In course of trading starting from a certain moment the message "Opening of Sell" will not be valid. In order not to misinform a user the graphical object must be deleted.

For deleting a graphical object (irrespective of its creation method - programmed or manual) simply select it and press the Delete key. However, as for programming, it should be noted that a correctly written program must "clear" the window when its operation is over. In other words, a program should contain a block where all graphical objects created by the program are deleted.

ObjectDelete() Function

bool ObjectDelete(string name)
deleting an object with the indicated name. If an object is successfully deleted, the function returns TRUE, otherwise - FALSE. To get the error information call the GetLastError() function..

Parameters:

· name - name of a deleted object. 

It is very easy to use ObjectDelete(): simply indicate the name of an object to delete.

To fix the disadvantage of the previous example, let us add into the EA grafobjects.mq4 the special function deinit() containing the function for deleting objects:

//--------------------------------------------------------------- 7 --
int deinit()                           // Special function deinit
  {
   ObjectDelete("Label_Obj_MACD");     // Object deletion
   return;                             // Exit deinit()
  }
//--------------------------------------------------------------- 8 --
Now, during the EA execution the object named Label_Obj_MACD will be deleted. Generally a program can create numerous objects. Each of them can be deleted according to the algorithm.

Modifying Graphical Objects

In some cases it is necessary to change an object position in a chart window in a program way. Very often such a necessity may occur because of the appearance of new bars. For example, trading criteria in an EA can be formed on the basis of a linear regression channel built on a bar history of a certain length (for example, last 50 bars). If we simply draw the object "linear regression channel" in a chart window and then do not undertake anything, it will remain on the same chart place where it was positioned and it will be shifted to the left as new bars appear. To prevent the object from shifting it should be redrawn at each new bar. For this purpose new coordinates must be calculated and passed to the object; in accordance with these coordinates the object will be drawn in a chart widow.

For finding out what properties a graphical object has at the current moment, the following function should be used:

ObjectGet() Function

double ObjectGet(string name, int prop_id)
the function returns the value of the specified object property. To get the error information call the GetLastError() function.

parameters:

· name - object name; 

· prop_id - object property identifier. Can be any value from the list of object properties. 

New coordinates are reported to an object using the ObjectMove() function.

ObjectMove() Function

bool ObjectMove(string name, int point, datetime time1, double price1)
Changing one of coordinates on a chart. The function returns TRUE in case of success, otherwise - FALSE. To get additional information call the FetLast Error() function. Numeration of an object coordinates starts from 0.

Parameters:

· name - object name; 

· point - coordinate index (0-2); 

· time1 - new time value; 

· price1 - new price value. 

	Problem 33. Create a program (an Expert Advisor) supporting the drawing of a linear regression channel for the last 50 bars.


The graphical object "linear regression channel" uses two time coordinates. Price coordinates (if such are specified in the program) are neglected by the client terminal during the object construction. The linear regression channel is calculated by the client terminal based on historic price data and therefore cannot be displayed aside from a chart. That is why the absence of the object binding to price (ignoring of price coordinates by the terminal) is the object's own constant property. Th Expert Advisor (moveobjects.mq4) managing the position of a graphical object can have the following code:

//--------------------------------------------------------------------
// moveobjects.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
extern int   Len_Cn=50;                // Channel length (bars)
extern color Col_Cn=Orange;            // Channel color
//--------------------------------------------------------------- 1 --
int init()                             // Special function init()
  {
   Create();                           // Calling user-def. func. of creation
   return;                             // Exit init()
  }
//--------------------------------------------------------------- 2 --
int start()                            // Special function start()
  {
   datetime T2;                        // Second time coordinate
   int Error;                          // Error code
//--------------------------------------------------------------- 3 --   
   T2=ObjectGet("Obj_Reg_Ch",OBJPROP_TIME2);// Requesting t2 coord.
   Error=GetLastError();               // Getting an error code
   if (Error==4202)                    // If no object :(
     {
      Alert("Regression channel is being managed",
            "\n Book_expert_82_2. deletion prohibited.");
      Create();                        // Calling user-def. func. of creation
      T2=Time[0];                      // Current value of t2 coordinate
     }
//--------------------------------------------------------------- 4 --
   if (T2!=Time[0])                    // If object is not in its place
     {
      ObjectMove("Obj_Reg_Ch", 0, Time[Len_Cn-1],0); //New t1 coord.
      ObjectMove("Obj_Reg_Ch", 1, Time[0],       0); //New t2 coord.
      WindowRedraw();                  // Redrawing the image 
     }
   return;                             // Exit start()
  }
//--------------------------------------------------------------- 5 --
int deinit()                           // Special function deinit()
  {
   ObjectDelete("Obj_Reg_Ch");         // Deleting the object
   return;                             // Exit deinit()
  }
//--------------------------------------------------------------- 6 --
int Create()                           // User-defined function..
  {                                    // ..of object creation
   datetime T1=Time[Len_Cn-1];         // Defining 1st time coord.
   datetime T2=Time[0];                // Defining 2nd time coord.
   ObjectCreate("Obj_Reg_Ch",OBJ_REGRESSION,0,T1,0,T2,0);// Creation
   ObjectSet(   "Obj_Reg_Ch", OBJPROP_COLOR, Col_Cn);    // Color
   ObjectSet(   "Obj_Reg_Ch", OBJPROP_RAY,   false);     // Ray
   ObjectSet(   "Obj_Reg_Ch", OBJPROP_STYLE, STYLE_DASH);// Style
   ObjectSetText("Obj_Reg_Ch","Created by the EA moveobjects",10);
   WindowRedraw();                     // Image redrawing
  }
//--------------------------------------------------------------- 7 --
The moveobjects.mq4 EA algorithm implies that an object attached once will remain on the screen during the whole time of the program execution. In such cases it is reasonable to use a user defined function (in this case it is Create(), block 6-7) for an object creation, the function can му called from the program anytime when needed. To draw an object two time coordinates are necessary (T1 is the coordinate of the object's left border, T2 - that of the right border):

   datetime T1 = Time[Len_Cn-1];       // Defining 1st time coord.
   datetime T2 = Time[0];              // Defining 2nd time coord.
In this example the right border of the object must always be on the zero bar, that is why the value of the second coordinate corresponds to the opening time of the zero bar. The left coordinate is calculated according to the number of bars set by a user (external variable Len_Cn) and is defined as the opening time of a bar with the corresponding index. For example, if the channel length is 50 bars, the left coordinate will be equal to the opening time of a bar with the index 49.

In the next lines of the user-defined function Create() the OBJ_REGRESSION object is created using ObjectCreate(), then necessary properties of the created object are set up by the ObjectSet() function (color preset by a user in an external variable, prohibited to draw as a ray, line style - dotted). In the line:

   ObjectSetText("Obj_Reg_Ch","Created by the EA moveobjects",10);
a text description is assigned to the object. As distinct from the earlier analyzed OBJ_LABEL, the text description of OBJ_REGRESSION is not displayed. The text description of graphical objects can be viewed in the object properties tab. This is very convenient in practical application for differentiating between objects created in a program way from those attached manually:

Fig. 138. Common properties of the graphical object "linear regression channel" created by the EA moveobjects.mq4.

Here is one more function used for redrawing of the current chart:

   WindowRedraw();                     // Image redrawing
WindowRedraw() Function

void WindowRedraw()
The function forcibly redraws the current chart. Usually it is used after object properties are changed.

Normally, the graphical objects are displayed by the client terminal in the sequence of incoming of new ticks. This is why, if we don't use WindowRedraw(), the changes in the object properties become visible to the user at the next tick, i.e., the displaying is always one tick late. The use of WindowRedraw() allows you to forcibly redraw all objects at a necessary moment, for example, immediately after the object properties have been changed. In a general case, if the properties of several objects are changed in the program, it is sufficient to use the function WindowRedraw() only once, after the properties of the last of the objects have been changed.

The user-defined function is first called from the special function init(). At the moment of attaching the EA to the symbol window, the execution of init() will start, which results in that the graphical object Linear Regression Channel will be displayed in the symbol window.

Two possible situations are considered in the function start(): (1) the object has been occasionally deleted by the user (block 3-4) and (2) it is necessary to move the object to the right when a new zero bar is formed (block 4-5). To detect whether the graphical object is available at the current moment, it is sufficient just to request the value of one of its coordinates. If the object exists, the function ObjectGet() will return a certain value that corresponds with the requested coordinate and the function GetLastError() will return zero value (i.e., no error occurred when requesting the coordinate). However, if there is no object of the given name in the symbol window, the function GetLastError() will return the code of error 4202, i.e., no object available:

   T2=ObjectGet("Obj_Reg_Ch",OBJPROP_TIME2);     // Requesting t2 coord.
   Error=GetLastError();                         // Getting an error code
If the error analysis showed that there were no object of that name, it means the program must create it, having notified the user about inadmissible actions (the program doesn't delete objects, it means that the object has been deleted by the user). This is why, after having displayed the message, the program calls to the previously considered user-defined function Create(), which results in a new creation of the object in the symbol window.

By the moment of the execution of the next block (4-5), the graphical object has already been created. To decide whether it must be moved, you should know the position of the object at the current moment. For this purpose, it is sufficient to analyze the previously obtained value of the first coordinate of the object. If this value doesn't coincide with the time of opening zero bar, to assign new coordinates to the object.

The coordinates are changed using the function ObjectMove():

   ObjectMove("Obj_Reg_Ch", 0, Time[Len_Cn-1],0); //New t1 coord.
   ObjectMove("Obj_Reg_Ch", 1, Time[0],       0); //New t1 coord.
Here, for the first coordinate (coordinate 0) of the object named Obj_Reg_Ch, the value of Time[Len_Cn-1] will be set, whereas for the second coordinate (coordinate 1) -Time[0]. The last parameters among those transferred to the function ObjectMove() is specified the parameter 0. This is the coordinate of the price that, according to the function description, must be transferred, but, in this case, will be ignored by the client terminal. As a result of execution of these lines, the properties of the considered graphical object will be changed. As a result of the next execution of the function WindowRedraw(), the graphical object will be forcibly redrawn by the client terminal - now according to the new values of the coordinates.

Thus, at the execution of the function start(), the graphical object Linear Regression Channel will be redrawn by the client terminal each time when a new bar forms, at its very first tick (see Fig. 139). After the execution of the EA has ended, the given graphical object will be deleted from the symbol window during execution of the special function deinit() (i.e., the program will "sweep out" its working place after the work has been finished).

Fig. 139. Displaying of the Linear Regression Channel at the execution of the EA moveobjects.mq4.

In a general case, you can create and delete graphical objects according to some conditions calculated in the program. You can display the support/resistance lines (OBJ_TREND), mark the time of approaching important events with vertical lines (OBJ_VLINE), indicate the intersections of various lines or the forecast price movements using text objects (OBJ_LABEL and OBJ_TEXT), etc.

It must be noted separately that, in a number of cases, there is no need to use graphical objects. For example, if you want to display in the screen a great variety of simple one-type images (for example, arrows), you can use indicator lines for this, having set their styles in the corresponding way. This approach will free you from the necessity to track many coordinates of objects in the program, it will also prevent you from occasional deletion of an image (the signs that display indicator lines can be neither selected nor deleted).

Functions for Working with Graphical Objects

	Function
	Summary Info

	ObjectCreate
	Creating an object with predefined name, type and initial coordinates in the indicated chart subwindow. number of object coordinates can be from 1 to 3 depending on the object type. In case of success the function returns TRUE, otherwise FALSE.

	ObjectDelete
	Deleting an object with the indicated name. In case of success the function returns TRUE, otherwise FALSE.

	ObjectDescription
	The function returns the object description. It returns for objects of the OBJ_TEXT and OBJ_LABEL types the text displayed in these objects.

	ObjectFind
	The function searches for the object of the given name. The function returns the index of the window, to which the searched object belongs. In case of failure, the function returns -1.

	ObjectGet
	The function returns the value of the given property of the object.

	ObjectGetFiboDescription
	The function returns the description of the Fibo object level. The amount of levels depends on the type of the object that belongs to the group of Fibo objects. The maximum amount of levels is 32.

	ObjectGetShiftByValue
	The functions calculates and returns the bar number (the shift relative to the current bar) for the given price. The bar number is calculated using a linear equation for the first and second coordinates. It is used for trend lines and similar objects.

	ObjectGetValueByShift
	The functions calculates and returns the price value for the given bar (the shift relative to the current bar). The price value is calculated using a linear equation for the first and second coordinates. It is used for trend lines and similar objects.

	ObjectMove
	Changing one of object coordinates on a chart. Objects can have from one to three anchoring points according to the object type. In case of success, the function returns TRUE, otherwise FALSE.

	ObjectName
	The function returns the object name according to its order number in the list of objects.

	ObjectsDeleteAll
	Deleting all object of the indicated type in the indicated chart subwindow. The function returns the number of deleted objects.

	ObjectSet
	Changing properties of an indicated object. In case of success the function returns TRUE, otherwise FALSE.

	ObjectSetFiboDescription
	The function assigns a new value to Fibonacci level. Number of levels depends on Fibonacci object type. Maximal number of levels is 32.

	ObjectSetText
	Changing object description. For objects OBJ_TEXT and OBJ_LABEL this description is displayed on a chart as a text line. In case of success the function returns TRUE, otherwise FALSE.

	ObjectsTotal
	Returns the total number of objects of the indicated type on a chart.

	ObjectType
	The function returns the type of an indicated object.


For the detailed description of these and other functions, please refer to Documentation at MQL4.community, MetaQuotes Software Corp. website or to "Help" section in MetaEditor.
Operations with Charts

In his or her practical work, a trader usually opens in a symbol window several subwindows that display indicators. There are no limitations on placing indicators, they can be attached in any sequence. The amount of subwindos in a symbol window is not limited either. Each subwindow has its number. The main window containing a price chart is always available, its number being 0. Each indicator subwindow has a number, as well. The subwindows are numbered in a simple sequence - they are numbered by their displaying in the symbol window from top to bottom: the indicator subwindow closest to the main window has number 1, the next lower one has number 2, the next one has number 3, etc.

Fig. 140. Subwindow locations in the symbol window.

The amount of subwindows can be easily calculated using the following function:

int WindowsTotal()
The function returns the amount of indicator subwindows located in the chart, including the main chart window. The largest number (of the lowest subwindow) is always 1 less than the total amount of subwindows (including the main window numbered as 0) in the symbol window. If, in the situation shown in Fig. 140, we call for execution the function WindowsTotal() from any application, the returned value will be equal to 3, while the largest number (of the lowest subwindow) is 2.

The numbering sequence described above is kept, if a new indicator subwindow is added to or an existing subwindow is deleted from the symbol window. If you add a new subwindow, it will be displayed below all other subwindows and its number is 1 more than that of the last window above it. If you delete a subwindow from the symbol window, all subwindows below it will be automatically renumbered - the number of each of them will be decreased by 1.

In MQL4, it is possible to create graphical objects (and change their properties) in any of the existing subwindows. For this purpose, in the function ObjectCreate() the parameter 'window' is provided, according to which an object is created in the given subwindow of the symbol window. The current number of the subwindow can be calculated using the following function:

int WindowFind(string name)
The function returns the number of the chart subwindow that contains the indicator named as 'name', if it has been found. Otherwise, it returns -1. The function will also return -1, if a custom indicator searches for itself during initialization init().

Parameters:

name - short name of the indicator.

The amount of subwindows in a symbol window can change at any moment, if the user deletes an indicator. This is why the algorithm of an application that supports monitoring of graphical objects must continuously track the numbers of windows, in which the indicators are displayed. 

	Problem 34. Using graphical objects, display the messages informing about data received from two indicators. If the corresponding indicator is attached to the symbol window, display the graphical object in the indicator window. Otherwise, display it in the main window.


To solve the problem, let's choose indicators RSI and Momentum. The general algorithm of building an Expert Advisor comes down to this. In the function init(), you can specify texts to be displayed on the screen according to the indicator readings, i.e., make the calculations to be executed only once in the program. In the function start(), you should calculate the indicator readings, detect the availability of necessary subwindows and their numbers, and then, according to the situation, display one or another message in one or another subwindow. At the execution of the function deinit(), it is necessary to delete all graphical objects created during the work of your program. Below is the EA named charts.mq4 that controls graphical objects in the subwindows of a symbol window.

//-----------------------------------------------------------------------------------
// charts.mq4
// The code should be used for educational purpose only.
//------------------------------------------------------------------------------ 1 --
int   Win_Mom_old=0,                          // Old number of subwindow Moment.
      Win_RSI_old=0;                          // Old number of subwindow RSI
color Color[5];                               // Declaration of the color array
string Text[5];                               // Declaration of the string array
//------------------------------------------------------------------------------ 2 --
int init()                                    // Special function init()
  {   
   Win_RSI_old=0;                             // Technical moment
   Win_Mom_old=0;                             // Technical moment
 
   Text[0]= "RSI(14) is below 30. Buy";       // Texts for situations RSI
   Text[1]= "RSI(14) is above 70. Sell";      // Texts for situations RSI
   Text[2]= "RSI(14) is between 30 and 70";   // Texts for situations RSI
   Text[3]= "Momentum(14) is growing";        // Texts for situations Momentum
   Text[4]= "Momentum(14) is sinking";        // Texts for situations Momentum
   Color[0]= DeepSkyBlue;                     // Object color for ..
   Color[1]= LightPink;                       // .. different situations ..
   Color[2]= Orange;                          // .. of the indicator RSI
   Color[3]= Color[0];                        // The same colors for Momentum
   Color[4]= Color[1];                        // The same colors for Momentum
 
   Create_RSI(0);                             // Creation of the first object
   Create_Mom(0);                             // Creation of the second object
   Main();                                    // Call to user-defined function
   return;                                    // Exit init()
  }
//------------------------------------------------------------------------------ 3 --
int start()                                   // Special function 'start'
  {
   Main();                                    // Call to the user-defined function
   return;                                    // Exit start()
  }
//------------------------------------------------------------------------------ 4 --
int deinit()                                  // Special function deinit()
  {
   ObjectDelete("Obj_RSI");                   // Deletion of the object
   ObjectDelete("Obj_Mom");                   // Deletion of the object
   return;                                    // Exit deinit()
  }
//------------------------------------------------------------------------------ 5 --
int Main()                                    // User-defined function
  {
   int                                        // Integer variables
   Win_RSI_new=0,                             // New number of the subwindow RSI
   Win_Mom_new=0,                             // New number of the subwindow Moment.
   Ind_RSI, Ind_Mom;                          // Indexes for situations
   double                                     // Real variables
   RSI,                                       // Value of RSI on bar 0
   Mom_0, Mom_1;                              // Value of Mom. on bars 0 and 1
//------------------------------------------------------------------------------ 6 --
   RSI=iRSI(NULL,0,14,PRICE_CLOSE,0);         // RSI(14) on zero bar
   Ind_RSI=2;                                 // RSI between levels 30 and 70
   if(RSI < 30)Ind_RSI=0;                     // RSI at the bottom. To buy
   if(RSI > 70)Ind_RSI=1;                     // RSI on the top. To sell     
//------------------------------------------------------------------------------ 7 --
   Win_RSI_new=WindowFind("RSI(14)");         // Window number of indicator RSI
   if(Win_RSI_new==-1) Win_RSI_new=0;         // If there is no ind., then the main window
   if(Win_RSI_new!=Win_RSI_old)               // Deleted or placed ..
     {                                        // .. window of indicator RSI
      ObjectDelete("Obj_RSI");                // Deletion of the object
      Create_RSI(Win_RSI_new);                // Create an object in the desired window
      Win_RSI_old=Win_RSI_new;                // Remember this window
     }                                        // Change the textual description:
   ObjectSetText("Obj_RSI",Text[Ind_RSI],10,"Arial",Color[Ind_RSI]);
//------------------------------------------------------------------------------ 8 --
   Mom_0=iMomentum(NULL,0,14,PRICE_CLOSE,0);  // Value on zero bar
   Mom_1=iMomentum(NULL,0,14,PRICE_CLOSE,1);  // Value on the preceding bar 
   if(Mom_0 >=Mom_1)Ind_Mom=3;                // Indicator line goes up
   if(Mom_0 < Mom_1)Ind_Mom=4;                // Indicator line goes down
//------------------------------------------------------------------------------ 9 --
   Win_Mom_new=WindowFind("Momentum(14)");    // Window number of indicator Momen
   if(Win_Mom_new==-1) Win_Mom_new=0;         // If there is no ind., then the main window
   if(Win_Mom_new!=Win_Mom_old)               // Deleted or placed ..
     {                                        // .. the window of Momentum indicator
      ObjectDelete("Obj_Mom");                // Deletion of the object
      Create_Mom(Win_Mom_new);                // Create an object in the desired window
      Win_Mom_old=Win_Mom_new;                // Remember this window
     }                                        // Change the textual description:
   ObjectSetText("Obj_Mom",Text[Ind_Mom],10,"Arial",Color[Ind_Mom]);
//----------------------------------------------------------------------------- 10 --
   WindowRedraw();                            // Redrawing the image 
   return;                                    // Exit the user-defined function
  }
//----------------------------------------------------------------------------- 11 --
int Create_RSI(int Win)                       // User-defined function
  {                                           // ..of creation of an object
   ObjectCreate("Obj_RSI",OBJ_LABEL, Win, 0,0); // Creation of an object
   ObjectSet("Obj_RSI", OBJPROP_CORNER, 0);     // Anchoring to an angle
   ObjectSet("Obj_RSI", OBJPROP_XDISTANCE, 3);  // Coordinate X
   if (Win==0)
      ObjectSet("Obj_RSI",OBJPROP_YDISTANCE,20);// Coordinate Y
   else
      ObjectSet("Obj_RSI",OBJPROP_YDISTANCE,15);// Coordinate Y
   return;                                      // Exit the user-defined function
  }
//----------------------------------------------------------------------------- 12 --
int Create_Mom(int Win)                         // User-defined function
  {                                             // ..creating an object
   ObjectCreate("Obj_Mom",OBJ_LABEL, Win, 0,0); // Creation of an object
   ObjectSet("Obj_Mom", OBJPROP_CORNER, 0);     // Anchoring to an angle
   ObjectSet("Obj_Mom", OBJPROP_XDISTANCE, 3);  // Coordinate X
   if (Win==0)
      ObjectSet("Obj_Mom",OBJPROP_YDISTANCE, 5);// Coordinate Y
   else
      ObjectSet("Obj_Mom",OBJPROP_YDISTANCE,15);// Coordinate Y
   return;                                      // Exit the user-defined function
  }
//----------------------------------------------------------------------------- 13 --
Prior to considering the code above, we should explain the specifics of the program operation. A graphical object once created (in this case, one displaying a text) is supposed to be present in the screen continuously. Its textual description is supposed to characterize the situation. The content of the textual description must be changed at the execution of the function start(), at every tick. At the same time, when switching between timeframes for the window, to which the EA is attached, the program goes through the following stages: deinit(), init(), (awaiting a tick), and start(). If the object is first time created during the execution of start(), then, every time when switching to another timeframe, a certain period of time will lapse before the object appears, the time period being equal to that of awaiting the next tick. This is very inconvenient, particularly, when timeframes are often switched between.

In a properly built program, the necessary messages are displayed on the screen at the moment of attaching the program to the symbol window or at the moment of timeframe switching (i.e., before a new tick incomes). For this purpose, as a rule, it is necessary to perform all actions to be performed at every tick at the launching of the special function start() at the stage of the execution of the special function init(). In order not to repeat the same program code in different special functions, the code can be organized as a separate function. For this purpose, the EA contains the user-defined function Main(). It is called to be executed once at the stage of initialization (block 2-3) and at every tick during the further work of the EA (block 3-4).

In the program (block 11-13), there are two more user-defined functions - Create_RSI() and Create_Mom() intended for creation and modification of the object properties. At the execution of the function init(), the necessary objects are created using these functions. The call to the function Main() results in giving necessary properties to the objects (the objects with the desired description of the desired color are displayed in the desired window).

Let's consider the function Main() (block 5-11) in more details. In block 6-7, the readings of the indicator RSI are calculated. Depending on whether the end of the indicator line is above 70, below 30, or within the range between these indexes, one or another value will be assigned to the variable Ind_RSI. Then this value is used as an index of arrays Color[] and Text[] (in block 7-8) to change the properties of the graphical object of the name "Obj_RSI".

Block 7-8. The number of RSI window is calculated in the line:

   Win_RSI_new = WindowFind("RSI(14)");// Window number of indicator RSI
The string value RSI(14) is used as the transferred parameter. This is the short name of the indicator, the number of which should be detected. In this case, the entire sequence of characters in the given line, including parentheses and digits, composes the name. It must be noted that, in general case, there can be several indicators of the same type in the symbol window, for example, RSI(14), RSI(21) and RSI(34). Each subwindow displaying these indicators has its own number. This is why technical indicators are developed in such a way that each of them forms its short name according to the preset values of adjustable parameters. The short name of each technical indicator coincides with that shown in the upper left corner of its subwindow (the short name of a custom indicator may be created by the programmer using the function IndicatorShortName()).

If the searched indicator is not placed in the symbol window, the variable Win_RSI_new (the number of the subwindow, in which this object must be displayed at the current moment) will take the value of -1, i.e., non-existing window. In this case, the program implies displaying of the graphical object in the main chart window, the number of which is always 0:**
   if(Win_RSI_new == -1) Win_RSI_new=0;// If there is no ind., then the main window
During his or her operations, the user may place a missing indicator or delete an existing one. In order to find out about what actions should be performed, the program uses global variables Win_RSI_old and Win_Mom_old. The value of each variable is the number of the subwindow, in which the object has previously been created. If the values of the variables Win_RSI_new and Win_RSI_old don't coincide, this means that the indicator window is either added (it didn't exist before) or deleted (it was available on the previous tick). In both cases, the previously created object must be deleted, and a new one must be created in the desired window:

      ObjectDelete("Obj_RSI");         // Deletion of the object
      Create_RSI(Win_RSI_new);         // Create an object in the desired window
After the object has been created in the window numbered as Win_RSI_new, the value equal to the number of this window is assigned to the variable Win_RSI_old, i.e., the program remembers the number of window, in which the graphical object was created:

      Win_RSI_old = Win_RSI_new;       // Remember this window
If the values of the variables Win_RSI_new and Win_RSI_old coincide, it means that it is sufficient to assign a textual description to the object (that is now placed in the necessary window). It must also be done, in case of creating a new object:

      ObjectSetText("Obj_RSI",Text[Ind_RSI],10,"Arial",Color[Ind_RSI]);
Similar calculations are made for the other subwindow, that of indicator Momentum (blocks 8 - 10). At the end of the function Main(), all available graphical objects are redrawn as a result of the execution of WindowRedraw().

It's easy to see that programmed control over graphical objects in subwindows implies using global variables (you can also use 'static'). In such cases, when coding a program, you should pay a special attention to what values can be taken by global variables in different situations and what this can result in. In the program considered above, global variables are zeroized in the function init():

   Win_RSI_old = 0;                    // Technical moment
   Win_Mom_old = 0;                    // Technical moment
These lines are included into the program due to the fact that global variables lose their values, only if the user has stopped the execution of the application program in the symbol window. However, if the user has adjusted external variables or switched the timeframe, the program undergoes deinitialization and the consequent initialization, the values of global variables being saved.

Let's consider the operations of the program that doesn't contain these lines. Suppose, both indicators with the subwindow numbers 1 and 2, respectively, have been placed in the symbol window by the moment when the user switches the timeframe. In the considered example, when deinitializing the program, the graphical objects are deleted. At the execution of the special function init(), the objects are created in zero window. Then, at the execution of the function Main(), in blocks 7-8 and 9-10, the program compares the obtained number of the window, in which the objects must be placed, and the number of the window, in which the objects were at the preceding tick. In fact, the object has already been placed in zero window, but the values of global variables will say for other result: their numbers will be 1 and 2. As a result, the graphical objects will remain in the main window, until the user deletes and reattaches the corresponding indicators. As to prevent these undesirable developments, the program implies nulling of global variables at the execution of the function init(). Thus, the values of these variables are made correspond with the situation.

As a result of the execution of EA charts.mq4, the following combinations of windows and graphical objects displayed can appear:

Fig. 141. Displaying graphical objects in the subwindows of a symbol window.

If there are both indicators in the symbol window, the corresponding graphical objects will be shown in their subwindows. If no one of these indicators is placed, then both objects will be created by the program in the main window. Adding or deletion of either indicator, the name of which is processed in the program, will result in moving of the corresponding graphical object into the necessary window. Adding or deletion of other indicators from the symbol window will not entail any consequences.

It must be noted separately that the alternative of deleting a graphical object by the user is not considered in this program. A program used in your practical trading must contain the analysis of such situation with the subsequent restoring of the object (see the solution of Problem 33).

Functions Used in Operations with Charts

	Function
	Summary Info

	HideTestIndicators
	The function puts a flag of hiding the indicators called by the Expert Advisor. At the opening the chart after testing, the indicators marked with the hiding flag will not be shown in the test graph. Before each call, the indicator is marked with the currently set hiding flag (only the indicators that are directly called from the EA under test can be displayed in the test graph).

	Period
	It returns the value of the amount of the period minutes for the current chart.

	RefreshRates
	Updating data in the predefined variables and timeseries arrays. This function is used, when an EA or a script has been calculating for a long time and needs updated data. It returns TRUE, if the data updating succeeds. Otherwise, it returns FALSE. The data may remain outdated only if they correspond with the current state of the client terminal.

	Symbol
	It returns a text line with the name of the current symbol.

	WindowBarsPerChart
	The function returns the amount of bars fitting in the window of the current chart.

	WindowExpertName
	It returns the name of the executing EA, script, custom indicator or library, depending on the MQL4 program, from which this function has been called. 

	WindowFind
	It returns the number of the chart subwindow that contains the indicator with the given name 'name' if it has been found. Otherwise, it returns -1. WindowFind() returns -1, if the custom indicator is searching for itself during initialization init().

	WindowFirstVisibleBar
	The function returns the number of the first visible bar in the window of the current chart. You should consider that price bars are numbered in a reversed order, from the last to the first one. The current bar, which is the last in the price array, has index 0. The oldest bar has index Bars-1. If the number of the first visible bar is 2 or more less than the amount of visible bars in the chart, this means that the chart window is not full and there is a space to the right.

	WindowHandle
	It returns the window handle for the window that contains the given chart. If no chart with symbol and timeframe is opened at the moment of function call, it returns 0.

	WindowIsVisible
	It returns TRUE, if the chart subwindow is visible. Otherwise, it returns FALSE. The chart subwindow chan be hidden due to the visibility properties of the indicator attached to it.

	WindowOnDropped
	It returns the index of window, in which the EA, a script or a custom indicator has been dropped. This value will be true, only if the EAs, custom indicators and scripts are attached using a mouse (the technology of 'drag and drop'). For custom indicators being initialized (call from the function init()), this index is not defined. The returned index is the number of the window (0 is the main chart window, indicator subwindows are numbered starting with 1), in which the custom indicator is working. During initialization, a custom indicator can create its won new subwindow, and its number will differ from that of the window, in which the indicator has really been dropped.

	WindowPriceMax
	It returns the maximum value of the vertical scale of the given subwindow of the current chart (0 is the main chart window, indicator subwindows are numbered starting with 1). If the subwindow index is not specified, the maximum value of the price scale of the main chart window will be returned.

	WindowPriceMin
	It returns the minimum value of the vertical scale of the given subwindow of the current chart (0 is the main chart window, indicator subwindows are numbered starting with 1). If the subwindow index is not specified, the minimum value of the price scale of the main chart window will be returned. 

	WindowPriceOnDropped
	It returns the price value at a chart point, at which an EA or a script have been dropped. The value will be true, only if the EA or the script have been moved using a mouse (the technology of 'drag and drop'). This value is not defined for custom indicators.

	WindowRedraw
	It redraws the current chart forcibly. The function is usually used after the object properties have been changed.

	WindowScreenShot
	It saves the display of the current chart in a GIF file. If it fails to make a screenshot, it returns FALSE.

	WindowTimeOnDropped
	The function returns the time value at a chart point, at which an EA or a script have been dropped. The value will be true, only if the EA or the script have been moved using a mouse (the technology of 'drag and drop'). This value is not defined for custom indicators.

	WindowsTotal
	The function returns the amount of indicator windows in the chart, including the main chart window. 

	WindowXOnDropped
	It returns the value of X coordinate in pixels for the point in the chart window client area, where an EA or a script have been dropped. The value will be true, only if the EA or the script have been moved using a mouse (the technology of 'drag and drop').

	WindowYOnDropped
	It returns the value of Y coordinate in pixels for the point in the chart window client area, where an EA or a script have been dropped. The value will be true, only if the EA or the script have been moved using a mouse (the technology of 'drag and drop').


For the detailed description of these and other functions, please refer to Documentation at MQL4.community, MetaQuotes Software Corp. website or to "Help" section in MetaEditor.

